

# Reading and Writing Arithmetic: Automating Ring Equalities in Agda

ANONYMOUS AUTHOR(S)

```

lemma : ∀ x y → x + y * 1 + 3 ≈ 2 + 1 + y + x

lemma x y = begin
  x + y * 1 + 3 ≈⟨ refl ⟨ +-cong ⟩ *-identityr y ⟨ +-cong ⟩ refl {3} ⟩
  x + y + 3   ≈⟨ +-comm x y ⟨ +-cong ⟩ refl ⟩
  y + x + 3   ≈⟨ +-comm (y + x) 3 ⟩
  3 + (y + x) ≈⟨ sym (+-assoc 3 y x) ⟩
  2 + 1 + y + x ■

```

lemma = solve NatRing

(a) A Tedious Proof                      (b) Our Solver

Fig. 1. Comparison Between A Manual Proof and The Automated Solver

We present a new library which automates the construction of equivalence proofs between polynomials over commutative rings and semirings in the programming language Agda [Norell and Chapman 2008]. It is significantly faster than Agda's existing solver. We use reflection to provide a simple interface to the solver, and demonstrate how to use the constructed relations to provide step-by-step solutions.

Additional Key Words and Phrases: proof automation, equivalence, proof by reflection, step-by-step solutions

CONTENTS				
		5.1.2	Sparse Encodings	8
		5.1.3	Hanging Indices	9
Abstract	1	5.2	Unification	10
Contents	1	5.2.1	Avoid Progress at all	
1 Introduction	1		Costs	10
2 Scope	2	5.2.2	Avoid Identities	11
3 Overview of the Proof Technique	3	5.3	Benchmarks	12
4 The Interface	4	6	Verification	13
4.1 Implementation	5	7	Pedagogical Solutions	13
4.2 Maintaining Invariants	6	8	Related Work	14
5 Performance	7	9	Conclusion	15
5.1 Normalisation	8		References	15
5.1.1 Horner Normal Form	8			

## 1 INTRODUCTION

Doing mathematics in dependently-typed programming languages like Agda has a reputation for being tedious. Even simple arithmetic identities like the one in Fig. 1 require fussy proofs (Fig. 1a).

This need not be the case! With the right components and libraries, we believe that Agda can be a tool for mathematics that's not only useful, but also easy, friendly, and fun. This work describes one such library: a solver which automates the construction of proofs like Fig. 1a, making them as easy as Fig. 1b.

The main contributions of our library are as follows:

**Ease of Use** Proofs like the one in Fig. 1a are long, difficult to write, and uninteresting. Our solver, in contrast, is extremely simple to use: the single line in Fig. 1b solves the lemma. This interface (section 4) is implemented using a lightweight reflection system (section 4.1), which doesn't require the user to write *any* reflection code, even if they use the solver with their own custom type.

**Performance** Our solver is significantly faster than Agda's current ring solver, cutting type-checking time down from minutes to seconds in several use cases (benchmarks in Fig. 7). As described in Grégoire and Mahboubi [2005], we use a sparse internal representation of polynomials (section 5.1.2). However, because of differences between Agda and Coq's type checker, we found that this optimisation—on its own—did not deliver a significant speedup, and actually damaged performance in a number of cases. Achieving the performance we did required an entirely separate kind of optimisation, described in section 5.2.

**Pedagogical Solutions** Computer algebra systems (CASs) outside the rigorous world of dependently-typed languages do far more than just check proofs for mistakes: they have a wealth of other features which can help with learning mathematics as well as verifying it. We hope that similar systems developed in Agda can do the same: as a demonstration, we implement “step-by-step solutions”, one of the most popular features of modern CASs. Far from being ill-suited to Agda, we show that the constructive nature of our proofs allows for a natural implementation (section 7).

## 2 SCOPE

In this section, we will explain the intended uses and limitations of the solver.

**Equivalences** First, an inflexibility: the solver deals very specifically with the domain of *equivalence* proofs, like the one in Fig. 1. While it may be of use in other settings (finding roots, etc.), that is not explored here.

**Setoids** On the other hand, we are very flexible about what kind of “equivalence” we prove. In fact, the solver will work with any equivalence relation, as long as it comes with proofs of the relevant ring axioms. This is useful for all of the usual things (approximating quotients and so on), but it also provides the basis for our “step-by-step solutions” implementation in section 7.

**Almost Rings** As in Grégoire and Mahboubi [2005, section 5], we use a peculiar algebraic structure which lies somewhere between a semiring and a ring. These “almost-rings” have all of the usual laws of a commutative ring, but instead of demanding additive inverses, they require the comparatively permissive “pseudo-inverse” operation, which obeys the following equations:

$$-(x * y) = -x * y \quad (1)$$

$$-(x + y) = -x + -y \quad (2)$$

This allows the solver to work on types which don't have an additive inverse (like  $\mathbb{N}$ ): such types just supply the identity function instead of negation, and the two laws above are satisfied.

A potential worry is that because we don't require  $x + -x = 0$  axiomatically, it won't be provable in our system. Happily, this is not the case: as long as  $1 + -1$  reduces to 0 in the coefficient set, the solver will verify the identity.

**Weak Decidability** A core optimisation in our solver (section 5.1.2) relies on the ability to test arbitrary coefficients for zero. Instead of requiring decidable equality (which would greatly




Fig. 2. The Reflexive Proof Process

diminish the number of types the solver can work with), we instead ask for weakly decidable equivalence with zero:

`is-zero :  $\forall x \rightarrow \text{Maybe } (0 \# \approx x)$`

Just as in Agda's current solver, this allows users to avail of the optimisation if their type supports it, or skip it (`is-zero = const nothing`) if not.

**Correctness** The nature of the solver means it is intrinsically sound (i.e. it cannot prove an equivalence unless there is one): since all it does is rearrange and join together the ring axioms, it cannot prove anything that doesn't derive from them. We have not, however, proven completeness (that every equivalence will be found by our solver).

In the internal representation of the solver, we prove several data structure invariants (like sparsity) intrinsically.

The reflection-based interface is unproven, but since the output is type checked our claim of soundness still stands: a bug in our reflection code can only cause the solver to miss a solution, never to prove something it shouldn't.

### 3 OVERVIEW OF THE PROOF TECHNIQUE

Before diving into specifics, we'll first give a quick overview of how the solver works, so it's clear how the bits of implementation described later in the paper fit together.

The technique we use for automating equivalence proofs comes from Boutin [1997]: the general idea is that we prove two expressions equivalent by proving that they're both equivalent to the same canonical form. The diagram in Fig. 2 demonstrates this for the identity from Fig. 1: on the bottom of the diagram you can see the left and right hand side of the identity we want to prove, and on the top we can see their normal forms. The actual proof the solver provides is represented by the  $\approx$  path.

To prove that each expression is equivalent to the canonical form we first represent the expressions using the type in Fig. 3. This type represents the Abstract Syntax Tree (AST) for expressions

```
data Expr {ℓ} (A : Set ℓ) (n : ℕ) : Set ℓ where
  K  : A → Expr A n
  I  : Fin n → Expr A n
  _⊕_ : Expr A n → Expr A n → Expr A n
  _⊗_ : Expr A n → Expr A n → Expr A n
  _⊗_ : Expr A n → ℕ → Expr A n
  _⊖_ : Expr A n → Expr A n
```

Fig. 3. A Type for Ring Expressions

```
lemma = +*-Solver.solve 2 (λ x y → x :+ y :* con 1 :+ con 3 := con 2 :+ con 1 :+ y :+ x) refl
```

Fig. 4. The Old Solver

in the almost-ring algebra: it has constructors corresponding to each ring operation ( $x + y = x \oplus y$ ,  $x * y = x \otimes y$ ), and it can refer to variables via their de Bruijn index (so  $x$  becomes  $! 0$ ).

The `[_]` function converts the AST to the expression we want to prove, whereas the `[_↓]` function converts expressions to a the canonical form. The implementation of the `[_↓]` function is described in section 5.1.

Equivalence of the canonical forms is proven via `correct`: some of the details of this are explained in section 6.

Finally, instead of asking the user to construct the AST themselves, we use reflection to automate it. This is described in the following section.

#### 4 THE INTERFACE

We felt an easy-to-use interface was one of the most important components of the library as a whole. Since we wanted to minimise the amount a user would have to learn to use the solver, we kept the surface area of the library quite small: aside from the `AlmostCommutativeRing` type described above, the rest of the interface consists of just two macros (`solve` and `solveOver`). We tried to make their usage as obvious as possible: just stick one of them (with the required arguments) in the place you need a proof, and the solver will do the rest for you.

`solve` is demonstrated in Fig. 1b. It takes a single argument: an implementation of the algebra. `solveOver` is designed to be used in conjunction with manual proofs, so that a programmer can automate a “boring” section of a larger more complex proof (Fig. 5). As well as the `AlmostCommutativeRing` implementation, this macro takes a list of free variables to use to compute the solution.

Because this interface is quite small, it’s worth pointing out what’s missing, or rather, what we *don’t* require from the user:

- We don’t ask the user to construct the `Expr` AST which represents their proof obligation. Compare this to Fig. 4: we had to write the type of the proof twice (once in the signature and again in the AST), and we had to learn the syntax for the solver’s AST. As well as being more verbose, this approach is less composable: every change to the proof type has to be accompanied by a corresponding change in the call to the solver. In contrast, the call to `solveOver` above effectively amounts to a demand for the compiler to “figure it out!” Any change to the expressions on either side will result in an *automatic* change to the proof constructed.
- We don’t ask the user to write any kind of “reflection logic” for their type. In other words, we don’t require a function which (for instance) recognises and parses the user’s type in the reflected AST, or a function which does the opposite, converting a concrete value into the AST that (when unquoted) would produce an expression equivalent to the quoted value.

```
lemma : ∀ x y → x + y * 1 + 3 ≈ 2 + 1 + y + x
lemma x y =
begin
  x + y * 1 + 3 ≈ { +comm (x + y * 1) 3 }
  3 + (x + y * 1) ≈ { solveOver (x :: y :: []) Nat.ring }
  3 + y + x      ≡ { }
  2 + 1 + y + x  ■
```

Fig. 5. The `solveOver` Macro

This kind of logic is complex, and very difficult to get right. While some libraries can assist with the task [Norell 2018; van der Walt and Swierstra 2013] it is still not fully automatic.

#### 4.1 Implementation

Agda has powerful metaprogramming facilities, which allow programs to manipulate their own code. Here, we'll use reflection to implement the interface to our solver.

Agda's reflection API is mostly encapsulated by the following three types:

**Term** The representation of Agda's AST, retrievable via `quoteTerm`.

**Name** The representation of identifiers, retrievable via `quote`.

**TC** The type-checker monad, which includes scoping and environment information, can raise type errors, unify variables, or provide fresh names. Computations in the **TC** monad can be run with `unquote`.

While `quote`, `quoteTerm`, and `unquote` provide all the functionality we need, they're somewhat low-level and noisy (syntactically speaking). Agda also provides a mechanism (which it calls macros) to package metaprogramming code so it looks like a normal function call (as in `solve`).

Reflection is obviously a powerful tool, but it has a reputation for being unsafe and error-prone. Agda's reflection system doesn't break type safety, but we *are* able to construct **Terms** which are ill-typed, which often result in confusing error-messages on the user's end. Unfortunately, constructing ill-typed terms is quite easy to do: the **Term** type itself doesn't contain a whole lot of type information, and it's quite fragile and sensitive to context. Variables, for instance, are referred to by their de Bruijn indices, meaning that the same **Term** can break if it's simply moved under a lambda.

Building a robust interface using reflection required a great deal of care. To demonstrate some of the techniques we used, we'll look at two functions from the core of the interface. First, `toExpr`:

```
toExpr : Term → Term
toExpr (def (quote AlmostCommutativeRing._+_ ) xs) = getBinOp (quote _⊕_) xs
toExpr (def (quote AlmostCommutativeRing._*_ ) xs) = getBinOp (quote _⊗_) xs
toExpr (def (quote AlmostCommutativeRing._^_ ) xs) = getExp xs
toExpr (def (quote AlmostCommutativeRing._-_ ) xs) = getUnOp (quote _⊖_) xs
toExpr v@(var x _) with x ℕ.<? numVars
... | yes p = v
... | no ¬p = constExpr v
toExpr t = constExpr t
```

This function is called on the **Term** representing one side of the target equivalence. It converts it to the corresponding **Expr**. In other words, it performs the following transformation:

$$x + y * 1 + 3 \rightarrow I\ 0 \oplus I\ 1 \otimes K\ 1 \oplus K\ 3$$

When it encounters one of the ring operators, it calls the corresponding helper function (`getBinOp`, `getExp`, or `getUnOp`) which finds the important subterms from the operator's argument list.

If it *doesn't* manage to match an operator or a variable, it assumes that what it has must be a constant, and wraps it up in the **K** constructor. This is the key trick which allows us to avoid ever asking the user to quote their own type. While it may seem unsafe at first glance, we actually found it to be more robust (for our use case) than the alternative:

*Principle 1* (Don't reimplement the typechecker). While it may seem good and fastidious to rigorously check the structure and types of arguments given to a macro, we found in general it was a

bad idea to do validity-checking in metaprogramming code. Instead, we preferred to proceed as if there were no errors (if possible), but arrange the output so that the user would still see a type error where the input was incorrect.

Taking this case as an example, if the user indeed manages to supply something other than the correct type, Agda will catch the error, as an incorrect argument to `K`.

If, on the other hand, we had asked the user to quote their own type, we would have trouble handling (for instance) closed applications of functions, references to names outside the lambda, etc. This approach, on the other hand, has no such difficulty.

Next, we'll look at one of the helper functions: `getExp`, which deals with exponentiation.

```
getExp : List (Arg Term) → Term
getExp (x <::> y <::> []) = quote _⊗_ { con } 3 ...<::> toExpr x <::> y <::> []
getExp (x :: xs) = getExp xs
getExp _ = unknown
```

It extracts the last two arguments to the exponentiation operator, and wraps them up with the `⊗`. Something worth noticing is that it doesn't care how long the list actually is, it only looks for the last two arguments. Again, we found that this looseness was actually a benefit:

*Principle 2* (Don't assume structure). In this example, as well as elsewhere, to find arguments to an  $n$ -ary function we simply extract the last  $n$  visible arguments to the function. While in theory we might be able to statically know all of the implicit and explicit arguments that will be used at the call-site, it's much simpler to ignore them, and try our best to be flexible. Remember, none of this is typed, so if something changes (like, say, a new universe level in [AlmostCommutativeRing](#)), you'll get type errors where you call `solve`, not where it's implemented.

The next thing to point out is `3 ...<::>`. This applies three hidden arguments as “unknown”, i.e. asks Agda to infer them. We could guess them ourselves: the first is the universe level of the carrier type, the second is the carrier type, and the third is the number of variables in the expression; however, we advise being as *inspecific* as possible, relying on the compiler to guess where you can:

*Principle 3* (Supply the minimal amount of information). There were several instances where, in constructing a term, we were tempted to supply explicitly some argument that Agda usually infers. Universe levels were a common example. In general, though, this is a bad idea: AST manipulation is fragile and error-prone, so the chances that you'll get some argument wrong are very high. Instead, you should *leverage* the compiler, relying on inference over direct metaprogramming as much as possible.

The final point to make is that the entire interface implementation is itself quite small (fewer than 100 lines). This isn't because our code was terse: rather, we intentionally minimised the amount of metaprogramming we did:

*Principle 4* (Try and implement as much of the logic outside of reflection as possible). With great power comes poor error messages, fragility, and a loss of first-class status. Therefore, If something can be done without reflection, *do it*, and use reflection as the glue to get from one standard representation to another.

## 4.2 Maintaining Invariants

One obvious benefit of reflection is a terse interface. However, we feel that another benefit—resilience to change—is just as important. This section illustrates that resilience with an example.

Agda allows us to encode program correctness in types, so we can *prove* properties we would have otherwise only been able to test. Unfortunately, these kinds of proofs tend to be very tightly coupled to the implementation of the algorithms they verify. This can make iteration difficult, where small optimisations or bug fixes can invalidate proofs for other invariants.

To demonstrate the problem, and how our solver can reduce some of the burden, we'll look at size-indexed binary trees:

```
data Tree : ℕ → Set a where
  leaf : Tree 0
  node : ∀ {n m} → A → Tree n → Tree m → Tree (n + m + 1)
```

We've deliberately chosen an awkward type here: in contrast to size-indexed lists, the index (the size) doesn't match the shape of the data structure. As a result, almost every function which manipulates the tree in some way will have to come accompanied by a verbose, complex proof.

Take this line, for instance, which performs a left-rotation on the tree:

```
rotl (node {a} x xl (node {b} {c} y yr yl)) = node y (node x xl yl) yr
```

A sensible invariant to encode here is that the function doesn't change the size of the tree. Unfortunately, to *prove* that invariant, we have to prove the following:

$$1 + (1 + a + c) + b = 1 + a + (1 + b + c)$$

Though simple, this is precisely the kind of proof which requires many fussy applications of the ring axioms. Here, our solver can help:

```
rotl (node {a} x xl (node {b} {c} y yr yl)) = node y (node x xl yl) yr
⇒ ∀ ( a :: b :: c :: [] )
```

While cutting down on the amount of code we need to write is always a good thing, the real strength of this method is that it automatically infers the input type. This makes it resilient to small changes in the code. So, when we notice the bug in the code above (*yl* and *yr* are swapped in the pattern-match), we can simply *fix it*, without having to touch any of the proof code.

```
rotl (node {a} x xl (node {b} {c} y yl yr)) = node y (node x xl yl) yr
⇒ ∀ ( a :: b :: c :: [] )
```

If we hadn't used the solver, this fix would have necessitated a totally new proof. By automating the proof, we allow the compiler to automatically check what we *mean* ("does the size of the tree stay the same?"), while we worry about other details.

## 5 PERFORMANCE

Type-checking proof-heavy Agda code is notoriously slow. Automated solvers like the one presented here can greatly exacerbate this cost: in our experience, it wasn't uncommon for Agda's current ring solver to spend several minutes proving a single identity.

Our solver is significantly faster than the current solver: the following sections will detail how we achieved that speedup. We will start by describing the naive implementation; in section 5.1.2 we demonstrate how we added the optimisations from Grégoire and Mahboubi [2005]; and in section 5.2 we will describe the Agda-specific optimisations which account for the bulk of our speedup. Finally, Fig. 7 contains some benchmarks against the current solver.



## 5.1 Normalisation

Most of code written for the solver is concerned with normalisation: the `[_↓]` function in Fig. 2. This converts from the expression AST (Fig. 3) to a canonical form.

**5.1.1 Horner Normal Form.** The particular “canonical form” we’ll start with is the same as in Agda’s current ring solver: Horner normal form. A polynomial (more specifically, a monomial) in  $x$  is represented as a list of coefficients of increasing powers of  $x$ . As an example, the following polynomial:

$$3 + 2x^2 + 4x^5 + 2x^7 \quad (3)$$

Is represented by this list:

$$3 :: 0 :: 2 :: 0 :: 4 :: 0 :: 2 :: []$$

Operations on these polynomials are similar to operations in positional number systems.

`_⊞_` : `Poly` → `Poly` → `Poly`

`[] ⊞ ys = ys`

`(x :: xs) ⊞ [] = x :: xs`

`(x :: xs) ⊞ (y :: ys) = x + y :: xs ⊞ ys`

`_⊗_` : `Poly` → `Poly` → `Poly`

`_⊗_ [] = []`

`_⊗_ (x :: xs) =`

`foldr (λ y ys → x * y :: map (_ * y) xs ⊞ ys) []`

So to get from `Expr` to `Poly` we map each constructor to the relevant polynomial operation. Then, to get from `Poly` to an expression in the underlying ring, we use Horner’s rule: a classic example of the `foldr` function.

`[_]` : `Poly` → `Carrier` → `Carrier`

`[ xs ] ρ = foldr (λ y ys → ρ * ys + y) 0# xs`

**5.1.2 Sparse Encodings.** Our first avenue for optimisation comes from Grégoire and Mahboubi [2005]. Notice that the encoding above is quite wasteful: it always stores an entry for each coefficient, even if it’s zero. In practice, we’re likely to often find long strings of zeroes (in expressions like  $x^{10}$ ), meaning that our representation will contain long “gaps” between the coefficients we’re actually interested in (non-zero ones).

To fix the problem we’ll switch to a *sparse* encoding, by storing a “power index” with every coefficient.

`Poly` : `Set c`

`Poly` = `List (Carrier × ℕ)`

This will represent the size of the gap from the previous non-zero coefficient. Taking 3 again as an example, we would now represent it as follows:

$$(3, 0) :: (2, 1) :: (4, 2) :: (2, 1) :: []$$

Next, we turn our attention to the task of adding multiple variables. Luckily, there’s an easy way to do it: nesting. Multivariate polynomials will be represented as “polynomials of polynomials”, where each level of nesting corresponds to one variable. It’s perhaps more clearly expressible in types:



```

393 Poly :  $\mathbb{N} \rightarrow \text{Set } c$ 
394 Poly zero = Carrier
395 Poly (suc n) = List (Poly  $n \times \mathbb{N}$ )

```

Inductively speaking, a “polynomial” in 0 variables is simply a constant, whereas a polynomial in  $n$  variables is a list of coefficients, which are themselves polynomials in  $n - 1$  variables.

Before running off to use this representation, though, we should notice that we have created another kind of “gap” which we should avoid with a sparse encoding. For a polynomial with  $n$  variables, we will always have  $n$  levels of nesting, even if the polynomial doesn’t actually refer to all  $n$  variables. In the extreme case, representing the constant 6 in a polynomial of 3 variables looks like the following:

```

404 ((((((6, 0) :: []), 0) :: []), 0) :: [])

```

The solution is another index: this time an “injection” index. This represents “how many variables to skip over before you get to the interesting stuff”. In contrast to the previous index, though, this one is type-relevant: we can’t just store a  $\mathbb{N}$  next to the nested polynomial to represent the gap. Because the polynomial is indexed by the number of variables it contains, any encoding of the gap will have provide the proper type information to respect that index.

**5.1.3 Hanging Indices.** The problem is a common one: we have a piece of code that works efficiently, and we now want to make it “more typed”, by adding more information to it, *without* changing the complexity class.

We found the following strategy to be useful: first, write the untyped version of the code, forgetting about the desired invariants as much as possible. Then, to add the extra type information, look for an inductive type which participates in the algorithm, and see if you can “hang” some new type indices off of it.

In our case, the injection index (distance to the next “interesting” polynomial) was simply stored as an  $\mathbb{N}$ , and the information we needed was the number of variables in the inner polynomial, and the number of variables in the outer. All of that is stored in the following proof of  $\leq$ :

```

422 data _≤_ (m :  $\mathbb{N}$ ) :  $\mathbb{N} \rightarrow \text{Set where}$ 
423   m≤m : m ≤ m
424   ≤-s  :  $\forall \{n\}$ 
425          $\rightarrow (m \leq n : m \leq n)$ 
426          $\rightarrow m \leq \text{suc } n$ 

```

A value of type  $n \leq m$  mimics the inductive structure of the  $\mathbb{N}$  we were storing to represent the distance between  $n$  and  $m$ . We were able to take this analogy quite far: in a few functions, for instance, we needed to compare these gaps. By mimicking the inductive structure of  $\mathbb{N}$ , we were able to directly translate **Ordering** and **compare** on  $\mathbb{N}$ :

```

432 data Ordering :  $\mathbb{N} \rightarrow \mathbb{N} \rightarrow \text{Set where}$ 
433   less  :  $\forall m k \rightarrow \text{Ordering } m (\text{suc } (m + k))$ 
434   equal :  $\forall m \rightarrow \text{Ordering } m m$ 
435   greater :  $\forall m k \rightarrow \text{Ordering } (\text{suc } (m + k)) m$ 

```

into equivalent functions on  $\leq$ :

```

442 data ≤-Ordering {n : ℕ} : ∀ {i j}                                → (i ≤ n : i ≤ n)
443                                     → (i ≤ n : i ≤ n)                → ≤-Ordering i ≤ n
444                                     → (j ≤ n : j ≤ n)                i ≤ n
445                                     → Set
446
447 where
448   ≤-lt : ∀ {i j-1}
449     → (i ≤ j-1 : i ≤ j-1)
450     → (j ≤ n : suc j-1 ≤ n)
451     → ≤-Ordering (≤-trans (≤-s i ≤ j-1) j ≤ n)
452                                     j ≤ n
453   ≤-gt : ∀ {i-1 j}
454     → (i ≤ n : suc i-1 ≤ n)
455     → (j ≤ i-1 : j ≤ i-1)
456     → ≤-Ordering i ≤ n
457                                     (≤-trans (≤-s j ≤ i-1) i ≤ n)
458   ≤-eq : ∀ {i}
459
460   ≤-compare : ∀ {i j n}
461     → (x : i ≤ n)
462     → (y : j ≤ n)
463     → ≤-Ordering x y
464
465   ≤-compare m ≤ m m ≤ m = ≤-eq m ≤ m
466   ≤-compare m ≤ m (≤-s y) = ≤-gt m ≤ m y
467   ≤-compare (≤-s x) m ≤ m = ≤-lt x m ≤ m
468   ≤-compare (≤-s x) (≤-s y)
469     with ≤-compare x y
470   ... | ≤-lt i ≤ j-1 _ = ≤-lt i ≤ j-1 (≤-s y)
471   ... | ≤-gt _ j ≤ i-1 = ≤-gt (≤-s x) j ≤ i-1
472   ... | ≤-eq _ = ≤-eq (≤-s x)

```

## 5.2 Unification

After applying the previous optimisations, we might expect an immediate speedup in the solver: unfortunately, this isn't the case. Without some careful adjustments, the optimisations in the previous section can actually *slow down* the solver. In this section, we'll try and explain the problem and how we fixed it, and give general guidelines on how to write Agda code which typechecks quickly.

Up until now, we have focused on the *operations* performed on the polynomial. Remember, though, the reflexive proof process has several steps: only one of them containing the operations ( $\llbracket \_ \rrbracket$  in Fig. 2). This isn't necessarily the most expensive step, though. Surprisingly, the innocuous-looking `refl` takes the bulk of the time! Typechecking this step involves unifying the two normalised expressions, a task which is quite expensive, with counterintuitive performance characteristics.

First, the good news. In the general case, unifying two expressions takes time proportional to the size of those expressions, so our hard-won optimisations do indeed help us.

Unfortunately, though, the "general case" isn't really that general: Agda's unification algorithm has a very important shortcut which we *must* make use of if we want our code to typecheck quickly: *syntactic equality*.

Before the full unification algorithm, Agda runs a quick check to see if the two expressions it's testing for equality are *syntactically* equal. This can make a big difference in unification problems like the following:

$$\text{sum } [1..100] \stackrel{?}{=} \text{sum } [1..100]$$

By noticing that these expressions are syntactically equal, we can avoid actually computing the `sum` function. Taking advantage of that shortcut is key to achieving decent performance. With that in mind, there are two main strategies we'll use to encourage syntactic equality:

**5.2.1 Avoid Progress at all Costs.** First, we will consider something which may seem inconsequential: the order of arguments to the evaluation functions.

$\llbracket xs \rrbracket_l \rho = \text{foldr } (\lambda y \ ys \rightarrow \rho * ys + y) \ 0 \ xs$        $\llbracket xs \rrbracket_r \rho = \text{foldr } (\lambda y \ ys \rightarrow y + ys * \rho) \ 0 \ xs$

$\llbracket \_ \rrbracket_l$  is the definition we've been working with so far. Some readers might find  $\llbracket \_ \rrbracket_r$  more natural, however. The reason is that it's more productive: in lazy languages, the usual convention is that functions which take multiple arguments should scrutinise those arguments from left to right. The  $*$  and  $+$  functions (on  $\mathbb{N}$ , at any rate) follow that convention, meaning that  $\llbracket \_ \rrbracket_r$  is able to make more progress without a concrete  $x$ . Taking the polynomial  $x^2 + 2$  as an example:

$\begin{aligned} \llbracket 2 :: 0 :: 1 :: \_ \rrbracket_l x &\equiv \langle \rangle \\ x * (x * (x * 0 + 1) + 0) + 2 &\equiv \langle \rangle \\ x * (x * (x * 0 + 1) + 0) + 2 &\blacksquare \end{aligned}$	$\begin{aligned} \llbracket 2 :: 0 :: 1 :: \_ \rrbracket_r x &\equiv \langle \rangle \\ 2 + (0 + (1 + 0 * x) * x) * x &\equiv \langle \rangle \\ \text{suc}(\text{suc}((x + 0) * x)) &\blacksquare \end{aligned}$
---	---

In  $\llbracket \_ \rrbracket_l$ , we're blocked pretty much straight away, as  $x$  is the first thing we try to scrutinise. In  $\llbracket \_ \rrbracket_r$ , since all of the constants are kept to the left, they're scrutinised first, allowing us to perform much more normalisation before being blocked.

Surprisingly, this is exactly what you *don't* want! Since both expressions will be coming out of the same evaluation function, they should have the same structure, meaning that we don't *need* the reduction of outer terms that  $\llbracket \_ \rrbracket_r$  gives us. We only need to perform normalisation on the coefficients: these are computed during the manipulations of the polynomial, and so may contain unevaluated expressions. If we used  $\llbracket \_ \rrbracket_r$  as our definition, then the type checker will likely hit an inequality on the *first* term, and as a result we lose all opportunity for syntactic equality.  $\llbracket \_ \rrbracket_l$ , on the other hand, front-loads all of the variables, maintaining syntactic equality for as long as possible.

As well as that, we don't have any control of the structure we get from the ring operators. This means that any reduction, as well as being unnecessary, can destroy the structural similarity between the two expressions, and as a result their syntactic equality.

The (counterintuitive) lesson learned is as follows: to speed up unification, keep things which are likely to be syntactically equal to the left, and *don't* structure your functions to encourage progress. Simply swapping the arguments (as we do above) resulted in a performance improvement of several orders of magnitude.

**5.2.2 Avoid Identities.** It's a good idea to avoid identities (expressions like  $0 + x$  or  $1 * x$ ) in the normalised expression. This will reduce the size of your expression, which is helpful in general, but more importantly it increases the likelihood of finding syntactic equality in the argument to the identity ( $x$  in the examples above).

Our sparse representation helps significantly in this case by entirely removing  $0$  from the generated expression. Another place we can cut down on constants is in identity elements in the base case for recursive functions. Take exponentiation, for example:

$$\begin{aligned} \_ \wedge \_ &: \text{Carrier} \rightarrow \mathbb{N} \rightarrow \text{Carrier} \\ x \wedge \text{zero} &= 1 \\ x \wedge \text{suc } i &= x * (x \wedge i) \end{aligned}$$

We can avoid that  $1$  in the majority of cases by rewriting the function to have an extra base case:

$\begin{aligned} \_ \wedge + 1 &: \text{Carrier} \rightarrow \mathbb{N} \rightarrow \text{Carrier} \\ x \wedge \text{zero} + 1 &= x \\ x \wedge \text{suc } i + 1 &= (x \wedge i + 1) * x \end{aligned}$	$\begin{aligned} \_ \wedge \_ &: \text{Carrier} \rightarrow \mathbb{N} \rightarrow \text{Carrier} \\ x \wedge \text{zero} &= 1 \\ x \wedge \text{suc } i &= x \wedge i + 1 \end{aligned}$
--	---




Fig. 6. Comparison of the normal forms of equation 3

In the library, we employ this idea extensively, avoiding unnecessary identities as much as we could. This has a significant effect on the size of the resulting normal form, but also ensures that normalisation stops exactly where we want it to, preserving the structure of the expressions as much as is possible. This makes a significant difference to both size and syntactic similarity as can be seen in Fig. 6.

### 5.3 Benchmarks

As expected, the new implementation exhibits a significant speedup in type-checking over the old implementation, even with the added overhead of the reflection-based interface. Fig. 7a shows time




Fig. 7. Time (in seconds) to prove each expression is equal to its expanded form ( $n = 5$  for each). Benchmarks performed on Agda version 2.6.0fa9b13, with the Agda standard library at commit-3bd3334a9552490e396f73f96812105a27e5917b, on a 2016 MacBook Pro, with a 2.9 GHz Intel Core i7 and 16 GB of RAM.

taken to type check a proof that  $(x_1 + x_2 + x_3 + x_4 + x_5)^d$  is equal to its expanded form. The new representation is clearly faster overall, with a factor of 7.5 speedup for  $d = 8$ .

Fig. 7c demonstrates perhaps a more common use-case, with a mix of high and low powers and some constants. The new representation's advantage is even more apparent here, with an 30-factor speedup at  $d = 8$ .

The old solver does exhibit a small lead (roughly 2-3 seconds, which narrows to about 1 second without reflection) on very simple expressions, possibly caused by the overhead of the new solver's more complex implementation. 3 seconds is quite small in the context of Agda type checking (the standard library, for instance, takes several minutes to type check), so we feel this slight loss is more than made up for by the several-minute gains. Furthermore, we have not been able to find a case where the old solver is significantly faster than the new. Nonetheless, if a user really wants to use the old solver, the other components described here are entirely modular, and can work with any underlying solver which uses the reflexive technique.

## 6 VERIFICATION

The output of the solver is a constructive proof of equivalence: this is *derived* from a generic proof that the operations on the solver are a ring homomorphism from the carrier type. Put another way, for the solver to work properly, we would need to prove that addition (and multiplication, and negation, etc.) on Horner normal forms corresponds with addition on the carrier type.

These proofs are long (about 1000 lines) and complex. Without careful structuring of the proofs, every new optimisation would require a whole new round of proof code, with very little reuse.

To avoid this problem, we took inspiration from [Mu et al. \[2009\]](#), and relied heavily on abstraction and folds to improve the reuse in proof code. In particular, we defined many operations as *metamorphisms* [\[Gibbons 2007\]](#). So, instead of defining (say) negation over the polynomial type itself, we will define a metamorphism to express negation, and then call some higher-order function to run that metamorphism over a polynomial.

$$\begin{aligned} \text{Meta} &: \mathbb{N} \rightarrow \text{Set } c \\ \text{Meta } n &= \text{Poly } n \times \text{Coeff } n \star \rightarrow \text{Poly } n \times \text{Coeff } n \star \end{aligned}$$

From here, we can define the *semantics* of a metamorphism. As an example, Fig. 8 shows the semantics of `poly-map`, a simple morphism which behaves something like `map` on lists.

Now, each operation only has to be proven up to the semantics defined above. Crucially, optimisations like the sparse encoding *respect* these semantics, so we only have to change our proof in one place: the definition of `poly-mapR`.

## 7 PEDAGOGICAL SOLUTIONS

One of the core aims of this work is to take a step towards making Agda a useful tool for doing mathematics. The rest of this paper has described our efforts to compensate for Agda's disadvantage in this area: namely, a pedantic typechecker. This section will attempt to show the other side of the coin, and demonstrate some of the unique

$$\begin{aligned} \text{poly-mapR} &: \forall \{n\} \rho \rho s \\ &\rightarrow ([f] : \text{Poly } n \rightarrow \text{Poly } n) \\ &\rightarrow (f : \text{Carrier} \rightarrow \text{Carrier}) \\ &\rightarrow (\forall x y \rightarrow x \star f y \approx f(x \star y)) \\ &\rightarrow (\forall x y \rightarrow f(x + y) \approx f x + f y) \\ &\rightarrow (\forall y \rightarrow ([f] y) \rho s \approx f([y] \rho s)) \\ &\rightarrow (f 0 \# \approx 0 \#) \\ &\rightarrow \forall xs \\ &\rightarrow \Sigma? [ \text{poly-map } [f] xs ] (\rho, \rho s) \\ &\approx f(\Sigma [ xs ] (\rho, \rho s)) \end{aligned}$$

Fig. 8. The Semantics of `poly-map`

benefits that come from using a programming language to do your proofs.

Outside of computer scientists and mathematicians, most people's experience of computer algebra probably amounts to the step-by-step solutions Wolfram|Alpha (or some similar system) provides online [The Development Team 2009].

Something so high-level and user-facing hardly seems like it's in a dependently-typed language's wheelhouse; on the other hand, the very nature of a proof in Agda is that it has computational content: why not make some of that content an explanation for the equality?

Prior work in this area includes Lioubartsev [2016]: there, the problem is reformulated reformulates the problem as one of *path-finding*. The left-hand-side and right-hand-side of the equation are vertices in a graph, where the edges are single steps to rewrite an expression to an equivalent form. A\* is used to search.

Unfortunately, this approach has to deal with a huge search space: every vertex will have an edge for almost every one of the ring axioms, and as such a good heuristic is essential. Furthermore, what this should be is not clear: Lioubartsev [2016] uses a measure of the “simplicity” of an expression.

Notice, however, that paths in undirected graphs form a perfectly reasonable equivalence relation: transitivity is the concatenation of paths, reflexivity is the empty path, and symmetry is *reversing* a path. Equivalence classes, in this analogy, are connected components of the graph.

More practically speaking, we implement these “paths” as lists, where the elements of the list are elementary ring axioms. When we want to display a step-by-step solution, we simply print out each element of the list in turn, interspersed with the states of the expression (the vertices in the graph).

If we stopped there, however, the solver would output incredibly verbose “solutions”: far too verbose to be human-readable. Instead, we must apply a number of path-compression heuristics to cut down on the solution length:

- (1) First, we remove loops from the graph. Fig 9 shows an example solution without this heuristic applied: it crosses the same point multiple times, creating the kind of loops we want to avoid. In contrast to using just A\* on its own, the search space is minimal (with only one outward edge for each vertex).
- (2) Then, we filter out “uninteresting” steps. These are steps which are obvious to a human, like associativity, or evaluation of closed terms. When a step is divided over two sides of an operator, it is deemed “interesting” if either side is interesting.




Fig. 9. Graph Containing Loops

After applying those heuristics, our solver outputs the explanation in Fig. 10 for the lemma in Fig. 1b.

## 8 RELATED WORK

```

x + y + 3
= { +-comm(x, y + 3) }
y + 3 + x
= { +-comm(y, 3) }
3 + y + x

```

Fig. 10. Step-by-Step Output From Our Solver

In dependently-typed programming languages, the state-of-the-art solver for polynomial equalities (over commutative rings) was originally presented in Grégoire and Mahboubi [2005], and is used in Coq's ring solver. This work improved on the already existing solver [Coq Development Team 2002] in both efficiency and flexibility. In both the old and improved solvers, a reflexive technique is used to automate the construction of the proof obligation (as described in Boutin [1997]).

Agda [Norell and Chapman 2008] is a dependently-typed programming language based on Martin-Löf’s Intuitionistic Type Theory [Martin-Löf 1980]. Its standard library [Danielsson 2018] currently contains a ring solver which is similar in flexibility to Coq’s ring, but doesn’t support the reflection-based interface, and is less efficient to the one presented here.

In Slama and Brady [2017], an implementation of an automated solver for the dependently-typed language Idris [Brady 2013] is described. The solver is implemented with a “correct-by-construction” approach, in contrast to Grégoire and Mahboubi [2005]. The solver is defined over *noncommutative* rings, meaning that it is more general (can work with more types) but less powerful (meaning it can prove fewer identities). It provides a reflection-based interface, but internally uses a dense representation.

Reflection and metaprogramming are relatively recent additions to Agda, but form an important part of the interfaces to automated proof procedures. Reflection in dependent types in general is explored in Christiansen [2015], and specific to Agda in van der Walt [2012].

Formalisation of mathematics in general is an ongoing project. Wiedijk [2018] tracks how much of “The 100 Greatest Theorems” [Kahl 2004] have so far been formalised (at time of writing, the number stands at 93). DoCon [Meshveliani 2018] is a notable Agda library in this regard: it contains many tools for basic maths, and implementations of several CAS algorithms. Its implementation is described in Meshveliani [2013]. Cheng et al. [2018] describes the manipulation of polynomials in both Haskell and Agda.

Finally, the study of *pedagogical* CASs which provide step-by-step solutions is explored in Lioubartsev [2016]. One of the most well-known such system is Wolfram Alpha [Wolfram Research, Inc. 2019], which has step-by-step solutions [The Development Team 2009].

## 9 CONCLUSION

We have presented an efficient, flexible, and easy-to-use ring solver in the programming language Agda. In doing so, we hope we have removed one of the small but significant pain points of doing maths in Agda.

## REFERENCES

- G Allais. 2011. Deciding Presburger Arithmetic Using Reflection. (May 2011). <https://gallais.github.io/pdf/presburger10.pdf>
- Samuel Boutin. 1997. Using Reflection to Build Efficient and Certified Decision Procedures. In *Theoretical Aspects of Computer Software (Lecture Notes in Computer Science)*, Martin Abadi and Takayasu Ito (Eds.). Springer Berlin Heidelberg, 515–529.
- Edwin Brady. 2013. Idris, a General-Purpose Dependently Typed Programming Language: Design and Implementation. *Journal of Functional Programming* 23, 05 (Sept. 2013), 552–593. <https://doi.org/10.1017/S095679681300018X>
- Chen-Mou Cheng, Ruey-Lin Hsu, and Shin-Cheng Mu. 2018. Functional Pearl: Folding Polynomials of Polynomials. In *Functional and Logic Programming (Lecture Notes in Computer Science)*. Springer, Cham, 68–83. [https://doi.org/10.1007/978-3-319-90686-7\\_5](https://doi.org/10.1007/978-3-319-90686-7_5)
- David Raymond Christiansen. 2015. *Practical Reflection and Metaprogramming for Dependent Types*. Ph.D. Dissertation. IT University of Copenhagen. <http://davidchristiansen.dk/david-christiansen-phd.pdf>
- The Coq Development Team. 2002. *The Coq Proof Assistant Reference Manual, Version 7.2*. <http://coq.inria.fr>
- Nils Anders Danielsson. 2018. The Agda Standard Library. <https://agda.github.io/agda-stdlib/README.html>
- Jeremy Gibbons. 2007. Metamorphisms: Streaming Representation-Change. *Science of Computer Programming* 65, 2 (2007), 108–139. <https://doi.org/10.1016/j.scico.2006.01.006>
- Benjamin Grégoire and Assia Mahboubi. 2005. Proving Equalities in a Commutative Ring Done Right in Coq. In *Theorem Proving in Higher Order Logics (Lecture Notes in Computer Science)*, Vol. 3603. Springer Berlin Heidelberg, Berlin, Heidelberg, 98–113. [https://doi.org/10.1007/11541868\\_7](https://doi.org/10.1007/11541868_7)
- Nathan W. Kahl. 2004. The Hundred Greatest Theorems. <http://web.archive.org/web/20080105074243/http://personal.stevens.edu/~nkahl/Top100Theorems.html>
- Pepijn Kokke and Wouter Swierstra. 2015. Auto in Agda. In *Mathematics of Program Construction (Lecture Notes in Computer Science)*, Ralf Hinze and Janis Voigtländer (Eds.). Springer International Publishing, 276–301. <http://www.staff.science>



uu.nl/~swier004/publications/2015-mpc.pdf

Dmitrij Lioubartsev. 2016. *Constructing a Computer Algebra System Capable of Generating Pedagogical Step-by-Step Solutions*. Ph.D. Dissertation. KTH Royal Institute of Technology, Stockholm, Sweden. <http://www.diva-portal.se/smash/get/diva2:945222/FULLTEXT01.pdf>

Per Martin-Löf. 1980. *Intuitionistic Type Theory*. Padua. <http://www.cse.chalmers.se/~peterd/papers/MartinL%00f6f1984.pdf>

Sergei D Meshveliani. 2013. *Dependent Types for an Adequate Programming of Algebra*. Technical Report. Program Systems Institute of Russian Academy of sciences, Pereslavl-Zalessky, Russia. 15 pages. <http://ceur-ws.org/Vol-1010/paper-05.pdf>

Sergei D. Meshveliani. 2018. DoCon-A a Provable Algebraic Domain Constructor. <http://www.botik.ru/pub/local/Mechveliani/docon-A/2.02/>

Shin-Cheng Mu, Hsiang-Shang Ko, and Patrik Jansson. 2009. Algebra of Programming in Agda: Dependent Types for Relational Program Derivation. *Journal of Functional Programming* 19, 5 (Sept. 2009), 545–579. <https://doi.org/10.1017/S0956796809007345>

Ulf Norell. 2018. Agda-Prelude: Programming Library for Agda. <https://github.com/UlfNorell/agda-prelude>

Ulf Norell and James Chapman. 2008. Dependently Typed Programming in Agda. (2008), 41.

Franck Slama and Edwin Brady. 2017. Automatically Proving Equivalence by Type-Safe Reflection. In *Intelligent Computer Mathematics*, Herman Geuvers, Matthew England, Osman Hasan, Florian Rabe, and Olaf Teschke (Eds.). Vol. 10383. Springer International Publishing, Cham, 40–55. [https://doi.org/10.1007/978-3-319-62075-6\\_4](https://doi.org/10.1007/978-3-319-62075-6_4)

The Development Team. 2009. Step-by-Step Math. <http://blog.wolframalpha.com/2009/12/01/step-by-step-math/>

Paul van der Walt and Wouter Swierstra. 2013. Engineering Proof by Reflection in Agda. In *Implementation and Application of Functional Languages*, Ralf Hinze (Ed.). Vol. 8241. Springer Berlin Heidelberg, Berlin, Heidelberg, 157–173. [https://doi.org/10.1007/978-3-642-41582-1\\_10](https://doi.org/10.1007/978-3-642-41582-1_10)

P. D. van der Walt. 2012. *Reflection in Agda*. Master’s Thesis. Universiteit of Utrecht. <https://dspace.library.uu.nl/handle/1874/256628>

Freek Wiedijk. 2018. Formalizing 100 Theorems. <http://www.cs.ru.nl/~freek/100/>

Wolfram Research, Inc. 2019. Wolfram|Alpha. Wolfram Research, Inc.. <https://www.wolframalpha.com/>