An Efficient and Flexible Evidence-Providing Polynomial Solver in Agda

D Oisín Kidney

September 13, 2018

Abstract	3	A Polynomial Solver	5
We present a new implementation of a ring solver in the programming language Agda [12]. The efficiency is improved over the version included in the standard library [3] by including optimizations described in [5], among others. We demonstrate techniques for constructing proofs based on the theory of lists, show how Agda's reflection system can be used to provide a safe and simple interface to the solver, and compare the "correct by construction" approach to that of auxiliary proofs. We also show that, as a by-product of proving equivalences rather than equalities, the generated proofs can be instantiated into a number of exotic settings, including: 1. Pretty-printing step-by-step solutions. 2. Providing human-readable counterexamples when a proof fails.	4 5 6	3.1 Choice of Algebra	5 6 6 7 7 8 9 9 9 9 10 111
3. Constructing isomorphisms between types represented as polynomials. Contents	7	Writing The Proofs 7.1 Equational Reasoning Techniques 7.2 The Algebra of Programming and List Homomorphisms	12 12
1 Introduction 2 2 A Case Study in Monoids 2 2.1 Equivalence Proofs 2 2.2 Canonical Forms 3 2.3 Homomorphism 4 2.4 Usage 5	8	Setoid Applications 8.1 Traced	12 12 13 13
2.5 Reflection	,	proach	13

1 Introduction

Dependently typed programming languages allow programmers and mathematicians alike to write proofs which can be executed. For programmers, this often means being able to formally verify the properties of their programs; for mathematicians, it provides a system of machine-checked verification not available to handwritten proofs.

Naïve usage of these systems can be tedious: the typechecker is often over-zealous in its rigor, demanding justification for every minute step in a proof, no matter how obvious or trivial it may seem to a human. For algebraic proofs, this kind of thing usually consists of long chains of rewrites, of the style "apply commutativity of +, then associativity of +, then at this position apply distributivity of * over +" and so on, when really the programmer wants to say "rearrange the expression into this form, checking it's correct".

However, since our proof assistant is also a programming language, we can provide the desired capability by writing a function which converts expressions into a canonical form, and a proof that the conversion preserves the semantics of the expression. This can then be used to automatically construct equivalence proofs for equivalent expressions.

2 A Case Study in Monoids

Before describing the ring solver, first we will explain the technique of writing a solver in Agda in the simpler setting of monoids.

Definition 2.1. Monoids A monoid is a set equipped with a binary operation, \bullet , and a distinguished element ϵ , such that the following equations hold:

$$x \bullet (y \bullet z) = (x \bullet y) \bullet z$$
 (Associativity)
 $x \bullet \epsilon = x$ (Left Identity)
 $\epsilon \bullet x = x$ (Right Identity)

Addition and multiplication (with 0 and 1 being the respective identity elements) are perhaps the most obvious instances of the algebra. In computer science, monoids have proved a useful abstraction for

```
record Monoid c \ \ell : Set (suc (c \sqcup \ell)) where infix f = 0 infix
```

Figure 1: The definition of Monoid in the Agda Standard Library [3]

formalizing concurrency (in a sense, an associative operator is one which can be evaluated in any order).

Monoids can be represented in Agda in a straightforward way, as a record (see Figure 1). Immediately it should be noted that we're no longer talking about a monoid over a set, but rather one over a setoid. In other words, rather than using propositional equality (indicated by the \equiv symbol), we will use a user-supplied equivalence relation (\approx in Figure 1) in our proofs.

2.1 Equivalence Proofs

Propositions are stated in type signatures in dependently typed languages. Figure 2 is an example of such a proposition.

```
ident: \forall w \ x \ y \ z

\rightarrow ((w \bullet \varepsilon) \bullet (x \bullet y)) \bullet z \approx (w \bullet x) \bullet (y \bullet z)
```

Figure 2: Example Identity

To a human, the fact that the identity holds may well be obvious: \bullet is associative, so scrub out all the parentheses, and ϵ is the identity element, so scrub it out too. After that, both sides are equal, so voilà!

Unfortunately, to convince the compiler we need to specify every instance of associativity and identity, rewriting the left-hand-side repeatedly until it matches the right:

```
ident w \ x \ y \ z =
begin
((w \bullet \varepsilon) \bullet (x \bullet y)) \bullet z
\langle (assoc \ (w \bullet \varepsilon) \ (x \bullet y) \ z \ )
(w \bullet \varepsilon) \bullet ((x \bullet y) \bullet z)
\langle (identity^r \ w \ (\bullet -cong \ ) \ assoc \ x \ y \ z \ )
w \bullet (x \bullet (y \bullet z))
\langle (x \bullet y) \bullet (y \bullet z) \ )
\langle (x \bullet y) \bullet (y \bullet z) \ )
\langle (x \bullet y) \bullet (y \bullet z) \ )
```

The syntax is designed to mimic that of a handwritten proof: line 3 is the expression on the left-hand side of \approx in the type, and line 9 the right-hand-side. In between, the expression is repeatedly rewritten into equivalent forms, with justification provided inside the angle brackets. For instance, to translate the expression from the form on line 3 to that on line 5, the associative property of \bullet is used on line 4.

Because we're not using propositional equality, some familiar tools are unavailable, like Agda's rewrite mechanism, or function congruence (this is why we have to explicitly specify the congruence we're using on line 6). The purpose of this particular hair shirt is flexibility: users can still use the solver even if their type only satisfies the monoid laws modulo some equivalence relation (perhaps they are have an implementation of finite, mergeable sets as balanced trees, and want to treat two sets as equivalent if their elements are equal, even if their internal structures are not). Beyond flexibility, we get some other interesting applications, which are explored in section 8.

Despite the pleasant syntax, the proof is mechanical, and it's clear that similar proofs would become tedious with more variables or more complex algebras (like rings). To avoid the tedium, then, we automate the procedure.

2.2 Canonical Forms

Automation of equality proofs like the one above can be accomplished by first rewriting both sides of the equation into a canonical form. Not every algebra has a canonical form: monoids do, though, and it's the simple list.

We're going to treat this type like an AST for a simple "language of lists". This language supports two functions: the empty list, and concatenation.

```
\begin{array}{l} \operatorname{infixr} 5 \ \_+ \_ \\ \_+ \_ : \ \forall \ \{i\} \rightarrow \operatorname{List} \ i \rightarrow \operatorname{List} \ i \rightarrow \operatorname{List} \ i \\ \boxed{] + ys = ys} \\ (x :: xs) \ + \ ys = x :: xs \ + \ ys \end{array}
```

The type itself parameterized by the number of variables it contains. Users can refer to variables by their index:

And we can interpret this language with values for each variable supplied in a vector:

```
\begin{array}{l} \_\mu\_: \ \forall \ \{i\} \rightarrow \mathsf{List} \ i \rightarrow \mathsf{Vec} \ \mathsf{Carrier} \ i \rightarrow \mathsf{Carrier} \\ [] \ \mu \ \rho = \varepsilon \\ (x:: xs) \ \mu \ \rho = \mathsf{lookup} \ x \ \rho \bullet xs \ \mu \ \rho \end{array}
```

Compare this language to the language of monoid expressions that Figure 2 uses: both have identity elements and a binary operator, and both refer to variables. Our language of lists, however, has one significant advantage, which can be seen when we rewrite the identity in Figure 2 in it:

```
obvious : (List 4 \ni ((\eta \# 0 \# []) \# (\eta \# 1 \# \eta \# 2)) \# \eta \# 3) \equiv (\eta \# 0 \# \eta \# 1) \# (\eta \# 2 \# \eta \# 3) obvious = \equiv .refl
```

The equivalence of each side is obvious to Agda, and so we don't need to write it out as we did before.

Proving that the language of lists and the language of monoid expressions¹ are equivalent is the task of the next section.

¹For simplicity's sake, from now on when we talk about

2.3 Homomorphism

Our obvious proof has the form:

$$lhs_{list} = rhs_{list} \tag{1}$$

What we want, though, is the following:

$$lhs_{mon} = rhs_{mon} \tag{2}$$

Equation 1 can be used to build Equation 2, if we supply two extra proofs:

$$lhs_{mon} \stackrel{a}{=} lhs_{list} = rhs_{list} \stackrel{b}{=} rhs_{mon}$$
 (3)

The proofs labeled a and b are the homomorphism proofs, and it's what we tackle next.

First, we'll define a concrete AST for the monoid language:

```
\begin{array}{ll} \mathsf{data} \ \mathsf{Expr} \ (i : \mathbb{N}) : \mathsf{Set} \ c \ \mathsf{where} \\ \_ \oplus \_ : \ \mathsf{Expr} \ i \to \mathsf{Expr} \ i \to \mathsf{Expr} \ i \\ \mathsf{e} \qquad : \ \mathsf{Expr} \ i \\ \mathsf{v} \qquad : \ \mathsf{Fin} \ i \to \mathsf{Expr} \ i \end{array}
```

It has constructors for each of the monoid operations $(\oplus$ and e are \bullet and ϵ , respectively), and it's indexed by the number of variables it contains, which are constructed with ν . Similarly to the list language, we can interpret it with a vector of values for the variables.

This interpreter will result in an unnormalized expression like the one in Figure 2; in other words, the outer sides of Equation 3. To get the normalized expressions, we'll first need to convert a monoid AST to a list AST:

the "language of monoid expressions", we won't be referring to curried functions of n arguments, rather we'll deal with functions which take a vector of length n, that refer to each variable by position. Translating between those two forms is not directly relevant to what we're doing here: we refer the interested reader to the Relation.Binary.Reflection module of Agda's standard library [3].

```
norm : \forall \{i\} \rightarrow \mathsf{Expr}\ i \rightarrow \mathsf{List}\ i

norm (x \oplus y) = \mathsf{norm}\ x + \mathsf{norm}\ y

norm e = []

norm (v \ x) = \eta \ x
```

Combining that with the fact that we know how to interpret the list AST, we get a normalizing interpreter for the monoid AST:

```
[\![\_\!\!\!\downarrow]\!\!]: \forall \{i\}
\rightarrow \mathsf{Expr}\ i
\rightarrow \mathsf{Vec}\ \mathsf{Carrier}\ i
\rightarrow \mathsf{Carrier}
[\![x \downarrow\!\!\!\downarrow]\!\!] \rho = \mathsf{norm}\ x \not\!\!\perp \rho
```

With this, we can see that the proofs needed by a and b are actually proofs of equivalence of these two interpreter functions.

Taking the non-normalizing interpreter as a template, the three cases are as follows²:

$$[x \oplus y] \rho \approx [x \oplus y \downarrow] \rho \tag{4}$$

$$[e]\rho \approx [e]\rho \approx (5)$$

$$[\![\forall i]\!] \rho \approx [\![\forall i \downarrow]\!] \rho \tag{6}$$

Proving each of these cases in turn finally verifies the correctness of our list language.

```
#-hom: \forall \{i\} (x \ y : \mathsf{List} \ i)
\rightarrow (\rho : \mathsf{Vec} \ \mathsf{Carrier} \ i)
\rightarrow (x + y) \ \mu \ \rho \approx x \ \mu \ \rho \bullet y \ \mu \ \rho
#-hom [] \ y \ \rho = \mathsf{sym} \ (\mathsf{identity}^{\mathsf{I}} \ \_)
#-hom (x :: xs) \ y \ \rho =
begin
\mathsf{lookup} \ x \ \rho \bullet (xs + y) \ \mu \ \rho
\approx (\mathsf{refl} \ ( \bullet \text{-cong} \ ) \ \text{+-hom} \ xs \ y \ \rho \ )
\mathsf{lookup} \ x \ \rho \bullet (xs \ \mu \ \rho \bullet y \ \mu \ \rho)
\approx (\mathsf{sym} \ (\mathsf{assoc} \ \_ \ \_ \ ) \ )
\mathsf{lookup} \ x \ \rho \bullet xs \ \mu \ \rho \bullet y \ \mu \ \rho

\bullet

correct: \forall \ \{i\}
\to (x : \mathsf{Expr} \ i)
\to (\rho : \mathsf{Vec} \ \mathsf{Carrier} \ i)
```

 $^{^2}$ Equations 4 and 5 comprise a monoid homomorphism.

2.4 Usage

Combining all of the components above, with some plumbing provided by the Relation.Binary.Reflection module, we can finally automate the solving of the original identity in figure 2:

```
 \begin{split} \mathsf{ident'} : \ \forall \ w \ x \ y \ z \\ & \ \rightarrow \ ((w \bullet \ \varepsilon) \bullet \ (x \bullet \ y)) \bullet \ z \\ & \ \approx \ (w \bullet x) \bullet \ (y \bullet z) \\ \mathsf{ident'} = \mathsf{solve} \ 4 \\ & (\lambda \ w \ x \ y \ z \\ & \ \rightarrow \ ((w \oplus \ e) \oplus \ (x \oplus y)) \oplus z \\ & \ \oplus \ (w \oplus x) \oplus \ (y \oplus z)) \\ \mathsf{refl} \end{aligned}
```

2.5 Reflection

One annoyance of the automated solver is that we have to write the expression we want to solve twice: once in the type signature, and again in the argument supplied to solve. Agda can infer the type signature:

```
ident-infer : \forall w \ x \ y \ z \rightarrow \_
ident-infer = solve 4
(\lambda w \ x \ y \ z
\rightarrow ((w \oplus e) \oplus (x \oplus y)) \oplus z
\stackrel{\textstyle (}{\oplus} (w \oplus x) \oplus (y \oplus z))
ref!
```

But we would prefer to write the expression in the type signature, and have it infer the argument to solve, as the expression in the type signature is the desired equality, and the argument to solve is something of an implementation detail.

This inference can be accomplished using Agda's reflection mechanisms.

Fill in reflection section

3 A Polynomial Solver

We now know the components required for an automatic solver for some algebra: a canonical form, a concrete representation of expressions, and a proof of correctness. We now turn our focus to polynomials.

Prior work in this area includes [14], [10], [16], [2], and [13], but perhaps the state-of-the-art (at least in terms of efficiency) is Coq's ring tactic [15], which is based on an implementation described in [5].

3.1 Choice of Algebra

The choice of algebra has been glossed over thus far, but it is an important design decision: choose one with too many laws, and the solver becomes unusable for several types; too few, and we may miss out on normalization opportunities.

The algebra defined in [5] is that of an almostring. It's very close in definition too a ring, but it discards the requirement that negation is an inverse (x + (-x) = 0). Instead, it merely requires that negation distribute over addition and multiplication appropriately. This allows the solver to be used with non-negative types, like \mathbb{N} , where negation is simply the identity function.

A potential worry is that because we don't require x+(-x)=0 axiomatically, it won't be provable in our system. This is not so: as long as 1-1 reduces to 0 in the coefficient set, the solver will verify the identity.

4 Horner Normal Form

The canonical representation of polynomials is a list of coefficients, least significant first ("Horner Normal Form"). Our initial attempt at encoding this representation will begin like so:

```
open import Algebra
```

```
module Dense \{\ell\} (coeff: RawRing \ell) where open RawRing coeff
```

The entire module is parameterized by the choice of coefficient. This coefficient should support the ring operations, but it is "raw", i.e. it doesn't prove the ring laws. The operations on the polynomial itself are defined like so³:

Finally, evaluation of the polynomial uses "Horner's rule" to minimize multiplications:

[_]: Poly
$$\rightarrow$$
 Carrier \rightarrow Carrier [x] ρ = foldr (λ y $ys \rightarrow y + \rho$ * ys) 0# x

4.1 Sparse Horner Normal Form

As it stands, the above representation has two prob-

- 1. Operators preceded with "N." are defined over N; e.g. N.+, N.*.
- 2. Plain operators, like + and $\star,$ are defined over the coefficients.
- 3. Boxed operators, like \boxplus and \boxtimes , are defined over polynomials
- 4. Operators which are boxed on one side are defined over polynomials on the corresponding side, and the coefficient on the other; e.g. ⋉, ⋊.

Redundancy The representation suffers from the problem of trailing zeroes. In other words, the polynomial 2x could be represented by any of the following:

$$0,2$$
 $0,2,0$
 $0,2,0,0$
 $0,2,0,0,0,0,0$

This is a problem for a solver: the whole *point* is that equivalent expressions are represented the same way.

Inefficiency Expressions will tend to have large gaps, full only of zeroes. Something like x^5 will be represented as a list with 6 elements, only the last one being of interest. Since addition is linear in the length of the list, and multiplication quadratic, this is a major concern.

In [5], the problem is addressed primarily from the efficiency perspective: they add a field for the "power index". For our case, we'll just store a list of pairs, where the second element of the pair is the power index⁴.

As an example, the polynomial:

$$3 + 2x^2 + 4x^5 + 2x^7$$

Will be represented as:

Or, mathematically:

$$x^{0}(3 + xx^{1}(2 + xx^{2} * (4 + xx^{1}(2 + x0))))$$

4.1.1 Uniqueness

While this form solves our efficiency problem, we still have redundant representations of the same polynomials. In [5], care is taken to ensure all operations

³Symbols chosen for operators use the following mnemonic:

⁴In [5], the expression (c,i):: P represents $P \times X^i + c$. We found that $X^i \times (c + X \times P)$ is a more natural translation, and it's what we use here. A power index of i in this representation is equivalent to a power index of i+1 in [5].

include a normalizing step, but this is not verified: in other words, it is not proven that the polynomials are always in normal form.

Expressing that a polynomial is in normal form turns out to be as simple as disallowing zeroes: without them, there can be no trailing zeroes, and all gaps must be represented by power indices. To check for zero, we require the user supply a decidable predicate on the coefficients. This changes the module declaration like so:

```
module Sparse \{a \ \ell\} (coeffs: RawRing \ a) (Zero: Pred (RawRing.Carrier \ coeffs) \ \ell) (zero?: Decidable \ Zero) where open RawRing coeffs
```

Finally, we can define a sparse encoding of Horner Normal Form:

```
infixl 6 = \neq 0

record Coeff : Set (a \sqcup \ell) where

inductive

constructor = \neq 0

field

coeff : Carrier

= \{ \text{coeff} \neq 0 \} : \neg Zero \text{ coeff} \}

open Coeff

Poly : Set (a \sqcup \ell)

Poly = List (Coeff \times \mathbb{N})
```

The proof of nonzero is marked irrelevant (preceded with a dot) to avoid computing it at runtime.

We can wrap up the implementation with a cleaner interface by providing a normalizing version of ::::

```
... | yes p = xs \triangle suc i
... | no \neg p = (\_ \neq 0 \ x \{ \neg p \} \ , \ i) :: xs
```

4.1.2 Comparison

Our addition and multiplication functions will need to properly deal with the new gapless formulation. First things first, we'll need a way to match the power indices. We can use a function from [8] to do so.

```
data Ordering : \mathbb{N} \to \mathbb{N} \to \mathsf{Set} where
  less
           : \forall m k
            \rightarrow Ordering m (suc (m \mathbb{N} + k))
  equal : \forall m
            \rightarrow Ordering m m
  \mathsf{greater}: \ \forall \ m \ k
            \rightarrow Ordering (suc (m \mathbb{N}.+ k)) m
compare : \forall m \ n \rightarrow \text{Ordering } m \ n
compare zero
                      zero
                                = equal zero
compare (suc m) zero
                                = greater zero m
                      (suc n) = less zero n
compare zero
compare (suc m) (suc n) with compare m n
compare (suc .m) (suc .(suc m \mathbb{N} + k))
  | less m k = less (suc m) k
compare (suc .m) (suc .m)
  | equal m = \text{equal (suc } m)
compare (suc .(suc m \mathbb{N} + k)) (suc .m)
  greater m k = \text{greater (suc } m) k
```

This is a classic example of a "leftist" function: after pattern matching on one of the constructors of Ordering, it gives you information on type variables to the *left* of the pattern. In other words, when you run the function on some variables, the result of the function will give you information on its arguments.

4.1.3 Efficiency

The implementation of **compare** may raise suspicion with regards to efficiency: if this encoding of polynomials improves time complexity by skipping the gaps, don't we lose all of that when we encode the gaps as Peano numbers?

The answer is a tentative no. Firstly, since we are comparing gaps, the complexity can be no larger than

that of the dense implementation. Secondly, the operations we're most concerned about are those on the underlying coefficient; and, indeed, this sparse encoding does reduce the number of those significantly. Thirdly, if a fast implementation of compare is really and truly demanded, there are tricks we can employ.

Agda has a number of built-in functions on the natural numbers: when applied to closed terms, these call to an implementation on Haskell's Integer type, rather than the unary implementation. For our uses, the functions of interest are -, +, <, and ==. The comparison functions provide booleans rather than evidence, but we can prove they correspond to the evidence-providing versions. Combined with judicious use of erase, we get the following:

```
less-hom : \forall n m
           \rightarrow ((n < m) \equiv \text{true})
           \rightarrow m \equiv \text{suc} (n + (m - n - 1))
less-hom zero zero ()
less-hom zero (suc m)
                            = refl
less-hom (suc n) zero ()
less-hom (suc n) (suc m) n < m =
  cong suc (less-hom n \ m \ n < m)
eq-hom : \forall n m
          \rightarrow ((n == m) \equiv \text{true})
          \rightarrow n \equiv m
eq-hom zero zero = refl
eq-hom zero (suc m) ()
eq-hom (suc n) zero ()
eq-hom (suc n) (suc m) n \equiv m =
  cong suc (eq-hom n m n \equiv m)
\mathsf{gt}	ext{-hom}: \ \forall \ n \ m
          \rightarrow ((n < m) \equiv false)
          \rightarrow ((n == m) \equiv false)
          \rightarrow n \equiv \text{suc} (m + (n - m - 1))
gt-hom zero zero n < m ()
gt-hom zero (suc m) () n \equiv m
gt-hom (suc n) zero n < m n \equiv m = \text{refl}
gt-hom (suc n) (suc m) n < m n \equiv m =
  cong suc (gt-hom n m n < m n \equiv m)
compare : (n m : \mathbb{N}) \to \text{Ordering } n m
compare n m with n < m | inspect (\_<\_ n) m
\dots | true | [ n < m ]
```

```
rewrite erase (less-hom n \ m \ n < m) =  less n \ (m - n - 1) ... | false | [n \not< m] with n == m | inspect (_==__ n) m ... | true | [n \equiv m] rewrite erase (eq-hom n \ m \ n \equiv m) = equal m ... | false | [n \not\equiv m] rewrite erase (gt-hom n \ m \ n \not< m \ n \not\equiv m) = greater m \ (n - m - 1)
```

4.1.4 Termination

Unfortunately, using compare in the most obvious way won't pass the termination checker.

Agda needs to be able to see that one of the numbers returned by compare always reduces in size: however, since the difference is immediately packed up in a list in the recursive call, it's buried too deeply in constructors for the termination checker to see it.

The solution is twofold: unpack any constructors into function arguments as soon as possible, and eliminate any redundant pattern matches in the offending functions. Taken together, these form an transformation known as "call pattern specialization" $[6]^5$. Happily, this transformation both makes termination more obvious and improves performance.

```
mutual infixl 6 _{\square} _{\square} _{\square} Poly \rightarrow Poly \rightarrow Poly
```

⁵This transformation is performed automatically by GHC as an optimization: perhaps a similar transformation could be performed by Agda's termination checker to reveal more terminating programs.

```
((x, i) :: xs) \boxplus ys = \coprod -zip-r \ x \ i \ xs \ ys
\boxplus-zip-r : Coeff \rightarrow \mathbb{N} \rightarrow \mathsf{Poly} \rightarrow \mathsf{Poly} \rightarrow \mathsf{Poly}
\boxplus-zip-r x \ i \ xs [] = (x , i) :: xs
\boxplus-zip-r x i xs ((y, j) :: ys) =
   \boxplus-zip (compare i j) x xs y ys
\boxplus-zip : \forall \{p \ q\}
         \rightarrow Ordering p q
         → Coeff
         → Poly
         → Coeff
         → Poly
         → Poly
\boxplus-zip (less i k) x xs y ys =
   (x, i) :: \boxplus -zip - r \ y \ k \ ys \ xs
\boxplus-zip (greater j k) x xs y ys =
   (y, j) :: \boxplus -zip - r \times k \times s ys
\boxplus-zip (equal i) x xs y ys =
   (coeff x + coeff y, i) :: \downarrow (xs \boxplus ys)
```

Every helper function in the mutual block matches on exactly one argument, eliminating redundancy. This also simplifies some of the homomorphism proofs later on.

5 Binary

Before continuing with polynomials, we'll take a short detour to look at binary numbers. These have a number of uses in dependently typed programming: as well as being a more efficient alternative to Peano numbers, their structure informs that of many data structures, such as binomial heaps, and as such they're used in proofs about those structures.

Similarly to polynomials, though, the naïve representation suffers from redundancy in the form of trailing zeroes. There are a number of ways to overcome this (see [9] and [4], for example); yet another is the repurposing of our sparse polynomial from above.

```
Bin: Set
Bin = List \mathbb{N}
```

is the only permitted coefficient. Effectively, all we store is the distance to another 1.

Addition (elided here for brevity) is linear in the number of bits, as expected, and multiplication takes full advantage of the sparse representation:

```
\mathsf{pow}:\,\mathbb{N}\to\mathsf{Bin}\to\mathsf{Bin}
pow i = 1
pow i(x :: xs) = (x \mathbb{N} + i) :: xs
pow x \circ \text{foldr} (\lambda \ y \ ys \rightarrow y :: xs + ys) []
```

6 Multivariate

Up until now our polynomial has been an expression in just one variable. For it to be truly useful, though, we'd like to be able to extend it to many: luckily there's a well-known isomorphism we can use to extend our earlier implementation. For a polynomial with two variables, we will represent it as a polynomial whose coefficients are themselves polynomials of one variable. For three, it'll be a polynomial whose coefficients are polynomials in two variables, and so on. Generally speaking, a multivariate polynomial is one where its coefficients are polynomials with one fewer variable [2].

Before jumping to implement this, though, we should notice an opportunity for optimization (also pointed out in [5]). Just like how the monomials had gaps of zeroes between non-zero terms, this type will have gaps of constant polynomials between nonconstant polynomials. In other words, in a expression of n variables, information about the last variable will be hidden behind n layers of nesting. If those n layers don't contain any information (i.e. they're all constant), we want to skip all that nesting with an injection index (like the power index).

6.1 Sparse Nesting

It's immediately clear that removing the gaps from We don't need to store any coefficients, because 1 the nesting will be more difficult than it was for the exponents: the Poly type is *indexed* by the number of variables it contains, so any manipulation of the injection index will have to correspond carefully to the Poly's index.

Our first approach might mimic the structure of Ordering, with an indexed type:

```
data Poly : \mathbb{N} \to \operatorname{Set} (a \sqcup \ell) where -\Pi_- : \forall i \{j\} \to \operatorname{FlatPoly} j \to \operatorname{Poly} (\operatorname{suc} (i \mathbb{N}.+j))
```

Where FlatPoly is effectively the gappy type we had earlier. If you actually tried to use this type, though, you'd run into issues: because it's an indexed type, pattern matching on it will force unification of the index with whatever type variable it was bound to. This is problematic because the index is defined by a function: pattern match on a pair of Polys and you're asking Agda to unify $i_1 + j_1$ and $i_2 + j_2$, a task it will likely find too difficult. How do we avoid this? "Don't touch the green slime!" [7]:

When combining prescriptive and descriptive indices, ensure both are in constructor form. Exclude defined functions which yield difficult unification problems.

We'll have to take another route.

6.1.1 Inequalities

First, we'll define our polynomial like so:

```
infixl 6 \ \Pi_record Poly (n : \mathbb{N}) : \mathsf{Set}\ (a \sqcup \ell) where inductive constructor \Pi_field \{i\} : \mathbb{N} flat : FlatPoly i i \le n : i \le n
```

The type is now parameterized, rather than indexed: our pattern-matching woes have been solved. Also, the gap is now implicit; instead, we store a proof that the nested polynomial has no more variables then the outer. Next, FlatPoly:

```
data FlatPoly : \mathbb{N} \to \operatorname{Set} (a \sqcup \ell) where \mathsf{K} : \operatorname{Carrier} \to \operatorname{FlatPoly} \operatorname{zero} \Sigma : \forall \{n\} \to (xs : \operatorname{Coeffs} n) \to .\{xn : \operatorname{Norm} xs\} \to \operatorname{FlatPoly} (\operatorname{suc} n)
```

We're back to an indexed type here, so you may be concerned about similar unification problems like the ones we had above: not to worry, though, as these indices are in constructor form, which prove much easier to unify than functions.

Also new here is the Norm function. It serves the same purpose that Zero did in the monomial, ensuring that there are no constant polynomials which could be replaced by incrementing the injection index. Its definition is as follows:

```
\begin{array}{lll} \mathsf{Norm} : \ \forall \ \{i\} \to \mathsf{Coeffs} \ i \to \mathsf{Set} \\ \mathsf{Norm} \ [] &= \bot \\ \mathsf{Norm} \ (\_ \ \Delta \ \mathsf{zero} & :: \ []) &= \bot \\ \mathsf{Norm} \ (\_ \ \Delta \ \mathsf{zero} & :: \ \_ :: \ \_) &= \top \\ \mathsf{Norm} \ (\_ \ \Delta \ \mathsf{suc} \ \_ :: \ \_) &= \top \end{array}
```

The rest of types are similar to what they were before:

```
infixl 6 \Delta
record CoeffExp (i:\mathbb{N}): Set (a\sqcup\ell) where
  inductive
  constructor \Delta
  field
     coeff: Coeff i
     pow : №
Coeffs: \mathbb{N} \to \mathsf{Set} (a \sqcup \ell)
Coeffs n = \text{List (CoeffExp } n)
infixl 6 #0
record Coeff (i : \mathbb{N}): Set (a \sqcup \ell) where
  inductive
  constructor #0
  field
     poly : Poly i
     .\{poly\neq 0\}: \neg Zero poly
Zero : \forall \{n\} \rightarrow \mathsf{Poly} \ n \rightarrow \mathsf{Set} \ \ell
```

```
 \begin{array}{lll} {\sf Zero} \; ({\sf K} \; x &  \  \, \Pi \; \_) = {\sf Zero} {\sf -C} \; x \\ {\sf Zero} \; (\Sigma \; [] &  \  \, \Pi \; \_) = {\sf Lift} \; \ell \; \top \\ {\sf Zero} \; (\Sigma \; (\_ :: \; \_) \; \Pi \; \_) = {\sf Lift} \; \ell \; \bot \\ \end{array}
```

Again, similarly to the sparse exponent encoding, we provide a smart constructor which ensures normalization.

6.1.2 Choosing an Inequality

Conspicuously missing above is a definition for \leq .

Option 1: The Standard Way The most commonly used definition of \leq is as follows:

```
\begin{array}{l} \mathsf{data} \ \_ \le \_ : \ \mathbb{N} \to \mathbb{N} \to \mathsf{Set} \ \mathsf{where} \\ \mathsf{z} \le \mathsf{n} : \ \forall \ \{n\} \to \mathsf{zero} \le n \\ \mathsf{s} \le \mathsf{s} : \ \forall \ \{m \ n\} \\ \ \to \ (m \le n : \ m \le n) \\ \ \to \ \mathsf{suc} \ m \le \mathsf{suc} \ n \end{array}
```

For our purposes, though, this type is dangerous: it actually *increases* the complexity from the dense encoding. To understand why, remember the addition function above with the gapless exponent encoding. For it to work, we needed to compare the gaps, and proceed based on that. We'll need to do a similar comparison on variable counts for this gapless encoding. However, we don't store the *gaps* now, we store the number of variables in the nested polynomial. Consider the following sequence of nestings:

$$(5 \le 6), (4 \le 5), (3 \le 4), (1 \le 3), (0 \le 1)$$

The outer polynomial has 6 variables, but it has a gap to its inner polynomial of 5, and so on. The comparisons will be made on 5, 4, 3, 1, and 0. Like repeatedly taking the length of the tail of a list, this is quadratic. There must be a better way.

Option 2: With Propositional Equality Once you realize we need to be comparing the gaps and not the tails, another encoding of \leq in Data. Nat seems the best option:

```
record \leq (m \ n : \mathbb{N}): Set where
constructor less-than-or-equal
field
\{k\} : \mathbb{N}
proof : m + k \equiv n
```

It stores the gap right there: in k!

Unfortunately, though, we're still stuck. While you can indeed run your comparison on k, you're not left with much information about the rest. Say, for instance, you find out that two respective ks are equal. What about the ms? Of course, you can show that they must be equal as well, but it requires a proof. Similarly in the less-than or greater-than cases: each time, you need to show that the information about k corresponds to information about k corresponds to information about k. Again, all of this can be done, but it all requires propositional proofs, which are messy, and slow. Erasure is an option, but I'm not sure of the correctness of that approach.

Option 3 What we really want is to *run* the comparison function on the gap, but get the result on the tail. Turns out we can do exactly that with the following:

```
\begin{array}{l} \text{infix 4} \ \_ \leq \_\\ \text{data} \ \_ \leq \_ \left( m : \mathbb{N} \right) : \mathbb{N} \ \rightarrow \ \text{Set where} \\ \text{m} \leq \text{m} : \ m \leq m \\ \leq \text{-s} \ : \ \forall \ \left\{ n \right\} \\ \ \rightarrow \ \left( m \leq n : \ m \leq n \right) \\ \ \rightarrow \ m \leq \ \text{suc} \ n \end{array}
```

While this structure stores the inequality by induction on the gap. That structure can be used to write a comparison function which was linear in the size of the gap (even though it compares the length of the tail).

6.1.3 Indexed Ordering

Now that the inequality is an inductive type, which mimics a Peano number stored in the gap, the parallels with the sparse exponent encoding should be even more clear. To write a comparison function, then, we Explain more the path from this to the final version. Intermediate comparison function and axiom K, especially

should first look for an equivalent to addition. This turns out to be transitivity:

```
\begin{array}{l} \text{infixl } 6 \ \_ \bowtie \_ \\ \_ \bowtie \_ : \ \forall \ \{x \ y \ z\} \rightarrow x \leq y \rightarrow y \leq z \rightarrow x \leq z \\ xs \bowtie \mathsf{m} \leq \mathsf{m} = xs \\ xs \bowtie (\leq -\mathsf{s} \ ys) = \leq -\mathsf{s} \ (xs \bowtie ys) \end{array}
```

With this defined, the Ordering type is obvious:

7 Writing The Proofs

The proofs are long (roughly 1000 lines), albeit mechanical.

7.1 Equational Reasoning Techniques

7.2 The Algebra of Programming and List Homomorphisms

8 Setoid Applications

I mentioned that the notion of equality we were using was more general than propositional, and that we could use it more flexibly in different contexts.

8.1 Traced

One "equivalence relation" is simply a labeled path: a list of rewrite rules or identities, repeatedly applied until the left-hand-side has been changed to the right. Print out the labels when done, and you have a step-by-step computer algebra system à la Wolfram Alpha. The definition of this type is straightforward:

```
infix 4 _≡···≡
infixr 5 \equiv \langle \rangle
data \_\equiv \dots \equiv \_: A \rightarrow A \rightarrow \mathsf{Set}\ a \ \mathsf{where}
    [refl]: \forall \{x\} \rightarrow x \equiv x
    \_\equiv \langle \_ \rangle \_ : \ \forall \ \{x\} \ y \ \{z\}
                          → String
                          \rightarrow y \equiv \cdots \equiv z
                          \rightarrow x \equiv \cdots \equiv z
    cong_1 : \forall \{x \ y \ z\} \{f : A \rightarrow A\}
                    → String
                   \rightarrow x \equiv \cdots \equiv y
                   \rightarrow f y \equiv \cdots \equiv z
                   \rightarrow f x \equiv \cdots \equiv z
    \mathsf{cong}_2: \ \forall \ \{x_1 \ x_2 \ y_1 \ y_2 \ z\} \ \{f: \ A \rightarrow A \rightarrow A\}
                   → String
                   \rightarrow x_1 \equiv \cdots \equiv x_2
                   \rightarrow y_1 \equiv \cdots \equiv y_2
                   \rightarrow f x_2 y_2 \equiv \cdots \equiv z
                   \rightarrow f x_1 \ y_1 \equiv \cdots \equiv z
```

And it does indeed implement the expected properties of an equivalence relation:

```
trans-\equiv \cdots \equiv : \ \forall \ \{x\ y\ z\}
\rightarrow x \equiv \cdots \equiv y
\rightarrow y \equiv \cdots \equiv z
\rightarrow x \equiv \cdots \equiv z

trans-\equiv \cdots \equiv [\text{refl}]\ ys = ys
trans-\equiv \cdots \equiv (y \equiv \langle\ x_1\ \rangle\ xs)\ ys =
y \equiv \langle\ x_1\ \rangle\ (\text{trans-}\equiv \cdots \equiv xs\ ys)
trans-\equiv \cdots \equiv (\text{cong}_1\ e\ x\equiv y\ fy\equiv z)\ ys =
\text{cong}_1\ e\ x\equiv y\ (\text{trans-}\equiv \cdots \equiv fy\equiv z\ ys)
trans-\equiv \cdots \equiv (\text{cong}_2\ e\ x\ y\ fxy\equiv z)\ ys =
\text{cong}_2\ e\ x\ y\ (\text{trans-}\equiv \cdots \equiv fxy\equiv z\ ys)
\text{cong}: \ \forall\ \{x\ y\}
\rightarrow (f:\ A\rightarrow A)
\rightarrow x\equiv \cdots \equiv y
```

Expand on the proofs. Operators used, etc.

actually been able to apply the "algebra of programming" [11] stuff in the proofs themselves yet. This section may well be removed if it turns out I

```
\rightarrow f x \equiv \cdots \equiv f y
cong f xs = cong_1 "cong" xs [refl]
\mathsf{sym}\text{-}\!\equiv\!\cdots\!\equiv:\;\forall\;\left\{x\;y\right\}\rightarrow x\equiv\!\cdots\!\equiv\;y\rightarrow\;y\equiv\!\cdots\!\equiv\;x
\operatorname{sym-=} \cdots \equiv \{x\} \ \{y\} = \operatorname{go} \ [\operatorname{refl}]
    where
        go: \forall \{z\}
               \rightarrow z \equiv \cdots \equiv x
               \rightarrow z \equiv \cdots \equiv y
               \rightarrow y \equiv \cdots \equiv x
        go xs [refl] = xs
        go xs (y \equiv \langle y? \rangle ys) =
             go ( \equiv \langle y? \rangle xs \rangle ys
        go xs (cong<sub>1</sub> e ys zs) =
            go (cong_1 \ e \ ys \ (\_ \equiv \langle \ e \ \rangle \ xs)) \ zs
        go xs (cong<sub>2</sub> e xp yp zp) =
            go (cong<sub>2</sub> e xp yp ( \equiv \langle e \rangle xs \rangle) zp
```

8.2 Isomorphisms

8.3 Counterexamples

9 The Correct-by-Construction Approach

The Agda and Coq communities exhibit something of a cultural difference when it comes to proving things. Coq users seem to prefer writing simpler, almost non-dependent code and algorithms, to separately prove properties about that code in auxiliary lemmas. Agda users, on the other hand, seem to prefer baking the properties into the definition of the types themselves, and writing the functions in such a way that they prove those properties as they go (the "correct-by-construction" approach).

There are advantages and disadvantages to each approach. The Coq approach, for instance, allows you to reuse the same functions in different settings, verifying different properties about them depending

on what's required. In Agda, this is more difficult: you usually need a new type for every invariant you maintain (lists, and then length-indexed lists, and then sorted lists, etc.). On the other hand, the proofs themselves often contain a lot of duplication of the logic in the implementation: in the Agda style, you avoid this duplication, by doing both at once. Also worth noting is that occasionally attempting to write a function that is correct by construction will lead to a much more elegant formulation of the original algorithm, or expose symmetries between the proof and implementation that would have been difficult to see otherwise.

[5], as an example, is very much in the Coq style: the definition of the polynomial type has no type indices, and makes no requirements on its internal structure:

```
Inductive Pol (C:Set) : Set :=
| Pc : C -> Pol C
| Pinj : positive -> Pol C -> Pol C
| PX : Pol C -> positive -> Pol C -> Pol C.
```

The implementation presented here straddles both camps: some properties, like the canonical form, or the number of variables in the polynomial, are baked into the type. The homomorphism properties, however, are proven in separate lemmas.

Performing the same task in a correct-by-construction way is explored in [14] (in Idris [1]).

Expand on the traced version, maybe clean it up? Also provide some examples.

Use the proof to translate between types. Check out Conor McBride's work on containers

Possible to provide counterexamples if a proof fails?

for this.

References

- [1] E. Brady, "Idris, a general-purpose dependently typed programming language: Design and implementation," *Journal of Functional Programming*, vol. 23, no. 05, pp. 552–593, Sep. 2013. [Online]. Available: http://journals.cambridge.org/article S095679681300018X
- [2] C.-M. Cheng, R.-L. Hsu, and S.-C. Mu, "Functional Pearl: Folding Polynomials of Polynomials," in Functional and Logic Programming, ser. Lecture Notes in Computer Science. Springer, Cham, May 2018, pp. 68–83. [Online]. Available: https://link.springer.com/chapter/ 10.1007/978-3-319-90686-7 5
- [3] N. A. Danielsson, "The Agda standard library," Jun. 2018. [Online]. Available: https://agda.github.io/agda-stdlib/README.html
- [4] M. Escardo, "Libraries for Bin," Jul. 2018.[Online]. Available: https://lists.chalmers.se/pipermail/agda/2018/010379.html
- [5] B. Grégoire and A. Mahboubi, "Proving Equalities in a Commutative Ring Done Right in Coq," in Theorem Proving in Higher Order Logics, ser. Lecture Notes in Computer Science, D. Hutchison, T. Kanade, J. Kittler, J. M. Kleinberg, F. Mattern, J. C. Mitchell, M. Naor, O. Nierstrasz, C. Pandu Rangan, B. Steffen, M. Sudan, D. Terzopoulos, D. Tygar, M. Y. Vardi, G. Weikum, J. Hurd, and T. Melham, Eds., vol. 3603. Berlin, Heidelberg: Springer Berlin Heidelberg, 2005, pp. 98–113. [Online]. Available: http://link.springer.com/10.1007/11541868
- [6] S. P. Jones, "Call-pattern specialisation for haskell programs." ACM Press, 2007, p. 327. [Online]. Available: https://www. microsoft.com/en-us/research/publication/ system-f-with-type-equality-coercions-2/

- [7] C. McBride, "A Polynomial Testing Principle,"
 Jul. 2018. [Online]. Available: https://twitter.com/pigworker/status/1013535783234473984
- [8] C. McBride and J. McKinna, "The View from the Left," J. Funct. Program., vol. 14, no. 1, pp. 69–111, Jan. 2004. [Online]. Available: http://strictlypositive.org/vfl.pdf
- [9] S. Meshveliani, "Binary-4 a Pure Binary Natural Number Arithmetic library for Agda," 21-Aug-2018. [Online]. Available: http://www. botik.ru/pub/local/Mechveliani/binNat/
- [10] S. D. Meshveliani, "Dependent Types for an Adequate Programming of Algebra," Program Systems Institute of Russian Academy of sciences, Pereslavl-Zalessky, Russia, Tech. Rep., 2013. [Online]. Available: http://ceur-ws.org/ Vol-1010/paper-05.pdf
- [11] S.-C. Mu, H.-S. Ko, and P. Jansson, "Algebra of programming in Agda: Dependent types for relational program derivation," *Journal of Functional Programming*, vol. 19, no. 5, pp. 545–579, Sep. 2009. [Online]. Available: https://www.cambridge.org/core/journals/journal-of-functional-programming/article/algebra-of-programming-in-agda-dependent-types-for-relational-ACA0C08F29621A892FB0C0B745254D15
- [12] U. Norell and J. Chapman, "Dependently Typed Programming in Agda," p. 41, 2008.
- [13] D. M. Russinoff, "Polynomial Terms and Sparse Horner Normal Form," Tech. Rep., Jul. 2017. [Online]. Available: http://www.russinoff.com/papers/shnf.pdf
- [14] F. Slama and E. Brady, "Automatically Proving Equivalence by Type-Safe Reflection," in Intelligent Computer Mathematics, H. Geuvers, M. England, O. Hasan, F. Rabe, and O. Teschke, Eds. Cham: Springer International Publishing, 2017, vol. 10383, pp. 40–55. [Online]. Available: http://link.springer.com/10.1007/978-3-319-62075-6_4

- [15] T. C. D. Team, "The Coq Proof Assistant, version 8.8.0," Apr. 2018. [Online]. Available: https://doi.org/10.5281/zenodo.1219885
- [16] U. Zalakain, "Evidence-providing problem solvers in Agda," Submitted for the Degree of B.Sc. in Computer Science, University of Strathclyde, Strathclyde, 2017. [Online]. Available: https://umazalakain.info/static/report.pdf